

# Hitachi HM55B Compass Module


#29123

PARALLAX

www.parallax.com  
support@parallax.com

## Out of the box

**Description:** The Hitachi HM55B Compass Module is a low cost, dual axis magnetic field sensor that can give your next electronic or robotic project a sense of direction. It is capable of detecting microtesla ( $\mu\text{T}$ ) variations in magnetic field strength. North/south and east/west sensing axes make it easy for the BASIC Stamp<sup>®</sup> microcontroller to resolve and display direction in a 360° format, with magnetic north at 0°, northeast at 45°, east at 90°, south at 180°, and so on.



A sense of direction opens up a variety of BASIC Stamp project and product possibilities. Here are just a few examples:

- Mobile robot direction sensor
- Weathervane indicator for remote weather stations
- Audible compass for the vision impaired
- Handheld electronic compass
- Automotive electronic compass

While this package insert provides an example circuit and test program, there are lots more online resources listed in the **Additional Resources** section.

### Features:


- Compact and breadboard-friendly package
- Easy to control and read with SHIFTOUT and SHIFIN
- Sensitive to microtesla ( $\mu\text{T}$ ) variations in magnetic fields
- Dual axes simplifies direction determination
- Built-in resistor protection for data pins
- 6-Bit (64-direction) or better resolution after calibration
- Only 30 to 40 ms between start measurement and data-ready

**Getting Started:** Build the circuit on your BASIC Stamp carrier board from the schematic shown below. Either hand enter the example program (see other side) into your BASIC Stamp Editor, or open it after downloading it from the product page listed in **Additional Resources**. Run the program and verify that the compass module gives you an angle measurement that corresponds to the direction it is pointing. If you are comparing the compass module's readings to a magnetic compass, use the magnetic compass to align a non-metallic object to the direction you want to test. Then, set the magnetic compass well away from the object before testing the compass module against it.

**For Best Results:** Go to the Hitachi HM55B Compass Module product page, and download the *Hitachi HM55B Compass Module Documentation.pdf*. Operate on flat surface, away from magnetic field disturbances such as large metal objects, motors, coils, magnets, power cords and programming cables.

**Precautions:** Do not operate or store near strong magnetic fields created by bar magnets, electromagnets, or electric motors.

## Schematic



## Additional Resources

Check out the Hitachi HM55B Compass Module product page for complete product documentation, example programs, the Hitachi<sup>®</sup> HM55B datasheet, and more:  
[http://www.parallax.com/detail.asp?product\\_id=29123](http://www.parallax.com/detail.asp?product_id=29123)

## Example BASIC Stamp Source Code

```
' -----[ Title ]-----
' TestHm55bCompass.bs2 - This Hitachi HM55B Compass Module test program displays x (N/S)
' and y (W/E) axis measurements along with the direction the Compass Module is pointing,
' measured in degrees clockwise from north.

' Author.... (C) 2005 Parallax, Inc -- All Rights Reserved
' Email..... support@parallax.com

' {$STAMP BS2}
' {$PBASIC 2.5}

' -----[ Pins/Constants/Variables ]-----
DinDout    PIN        6           ' P6 traneives to/from Din/Dout
Clk        PIN        5           ' P5 sends pulses to HM55B's Clk
En         PIN        4           ' P4 controls HM55B's /EN(ABLE)

Reset      CON        %0000      ' Reset command for HM55B
Measure    CON        %1000      ' Start measurement command
Report     CON        %1100      ' Get status/axis values command
Ready      CON        %1100      ' 11 -> Done, 00 -> no errors
NegMask    CON        %1111100000000000 ' For 11-bit negative to 16-bits

x          VAR        Word       ' x-axis data
y          VAR        Word       ' y-axis data
status     VAR        Nib        ' Status flags
angle      VAR        Word       ' Store angle measurement

' -----[ Main Routine ]-----
DO                                     ' Main loop

  GOSUB Compass_Get_Axes              ' Get x, and y values

  angle = x ATN -y                    ' Convert x and y to brads
  angle = angle */ 360                ' Convert brads to degrees

  DEBUG HOME,                          ' Display axes and degrees
    "x-axis N(-S) = ",SDEC x,
    CLREOL, CR, "y-axis W(-E) = ",
    SDEC y, CLREOL, CR, CR, "angle = ",
    DEC angle, " Degrees", CLREOL

  PAUSE 150                            ' Debug delay for slower PCs

LOOP                                   ' Repeat main loop

' -----[ Subroutines ]-----
Compass_Get_Axes:                     ' Compass module subroutine

  HIGH En: LOW En                      ' Send reset command to HM55B
  SHIFTOUT DinDout,clk,MSBFIRST,[Reset\4]

  HIGH En: LOW En                      ' HM55B start measurement command
  SHIFTOUT DinDout,clk,MSBFIRST,[Measure\4]
  status = 0                            ' Clear previous status flags

  DO                                    ' Status flag checking loop
    HIGH En: LOW En                    ' Measurement status command
    SHIFTOUT DinDout,clk,MSBFIRST,[Report\4]
    SHIFTIM DinDout,clk,MSBPOST,[Status\4]
  LOOP UNTIL status = Ready            ' Exit loop when status is ready
  SHIFTIM DinDout,clk,MSBPOST,[x\11,y\11]
  HIGH En                               ' Disable module

  IF (y.BIT10 = 1) THEN y = y | NegMask ' Store 11-bits as signed word
  IF (x.BIT10 = 1) THEN x = x | NegMask ' Repeat for other axis

RETURN
```